The Gramm-Schmidt Process

Worksheet by Joyce Quella -Winona State University (jquella@winona.edu)

Gill Tepper -Howard University (tgill@howard.edu)

Marius Buliga -Univ. of Pittsburgh-Bradford (buliga@pitt.edu)

```
> restart: with(LinearAlgebra): with(plots): with(plottools):
Warning, the name changecoords has been redefined
Warning, the assigned name arrow now has a global binding
```

Outline

The basic objectives are:

1) Learn the basic mechanics of the Gramm-Schmidt process for sets of vectors in the Euclidean spaces R^2 and R^3. Visualize the initial set of vectors and the ortogonal/orthonormal sets.

2) Write a Maple procedure that implements the Gramm-Schmidt process explored in Part 1 for larger sets of vectors from R^n . Run the procedure to check the results from Part 1.

3) Implement the Gramm-Schmidt process for a non-Euclidean space.

```
>
>
```

Gramm-Schmidt Process for Sets of Vectors in R^2 and R^3:

We start with a nonorthogonal basis in R^2:

```
> u1 := <1, 1>; u2 := <1, 3>;
```

$$u1 := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

$$u2 := \begin{bmatrix} 1 \\ 3 \end{bmatrix} \quad (2.1)$$

We apply the Gramm-Schmidt Process to orthogonalize this base:

```
> ogu1:=u1;
  ogu2:=u2-DotProduct(u2,ogu1)/
           DotProduct(ogu1,ogu1)*ogu1;
```

$$ogu1 := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

$$ogu2 := \begin{bmatrix} -1 \\ 1 \end{bmatrix} \quad (2.2)$$

It is easy to check that the two new vectors are orthogonal by computing their dot product:

```
> DotProduct(ogu1,ogu2);
```

$$0 \tag{2.3}$$

If we want to orthonormalize them, we have to divide each of them by their norm:

```
> onu1:=1/Norm(ogu1,2)*ogu1;
  onu2:=1/Norm(ogu2,2)*ogu2;
```

$$onu1 := \begin{bmatrix} \frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} \end{bmatrix}$$

$$onu2 := \begin{bmatrix} -\frac{1}{2}\sqrt{2} \\ \frac{1}{2}\sqrt{2} \end{bmatrix} \tag{2.4}$$

We'll plot the initial set of vectors {u1,u2} and the orthonormal set {onu1,onu2}.

```
> plu1 := plots[arrow](u1, shape=arrow,color=green,thickness=2):
  plu2 := plots[arrow](u2, shape=arrow,color=green,thickness=2):
  plonu1 := plots[arrow](onu1, shape=arrow,color=red,thickness=5)
  :
  plonu2 := plots[arrow](onu2, shape=arrow,color=red,thickness=5)
  :
  display(plu1,plu2,plonu1,plonu2, scaling=CONSTRAINED, axes=
  FRAMED);
>
```



Note that onu1 is collinear with u1.

Note that Maple has also the command GramSchmidt:

```
> GramSchmidt([u1,u2]);
```

$$\left[\begin{bmatrix}1\\1\end{bmatrix}, \begin{bmatrix}-1\\1\end{bmatrix}\right] \tag{2.5}$$

```
> GramSchmidt([u1,u2],normalized);
```

$$\left[\begin{bmatrix}\frac{1}{2}\sqrt{2}\\ \frac{1}{2}\sqrt{2}\end{bmatrix}, \begin{bmatrix}-\frac{1}{2}\sqrt{2}\\ \frac{1}{2}\sqrt{2}\end{bmatrix}\right] \tag{2.6}$$

We try next an example in R^3. We take an arbitrary base:

```
> v1 := <1, 1,1>; v2 := <-1, 3,2>; v3 := <-1, 0,5>;
```

$$v1 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$v2 := \begin{bmatrix} -1 \\ 3 \\ 2 \end{bmatrix}$$

$$v3 := \begin{bmatrix} -1 \\ 0 \\ 5 \end{bmatrix} \qquad (2.7)$$

We apply the Gramm-Schmidt Process to orthogonalize this base:

```
> ogv1:=v1;
  ogv2:=v2-DotProduct(v2,ogv1)/
           DotProduct(ogv1,ogv1)*ogv1;
  ogv3:=v3-DotProduct(v3,ogv1)/
           DotProduct(ogv1,ogv1)*ogv1
          -DotProduct(v3,ogv2)/
           DotProduct(ogv2,ogv2)*ogv2;
```

$$ogv1 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$ogv2 := \begin{bmatrix} \frac{-7}{3} \\ \frac{5}{3} \\ \frac{2}{3} \end{bmatrix}$$

$$ogv3 := \begin{bmatrix} \frac{-21}{26} \\ \frac{-63}{26} \\ \frac{42}{13} \end{bmatrix} \qquad (2.8)$$

We check that the three new vectors are orthogonal by computing their dot product:

```
> DotProduct(ogv1,ogv2);
  DotProduct(ogv1,ogv3);
  DotProduct(ogv2,ogv3);
```

$$0$$

$$0$$

$$0 \qquad (2.9)$$

We orthonormalize them:

```
> onv1:=1/Norm(ogv1,2)*ogv1;
  onv2:=1/Norm(ogv2,2)*ogv2;
  onv3:=1/Norm(ogv3,2)*ogv3;
```

$$onv1 := \begin{bmatrix} \frac{1}{3}\sqrt{3} \\ \frac{1}{3}\sqrt{3} \\ \frac{1}{3}\sqrt{3} \end{bmatrix}$$

$$onv2 := \begin{bmatrix} -\frac{7}{78}\sqrt{78} \\ \frac{5}{78}\sqrt{78} \\ \frac{1}{39}\sqrt{78} \end{bmatrix}$$

$$onv3 := \begin{bmatrix} -\frac{1}{26}\sqrt{26} \\ -\frac{3}{26}\sqrt{26} \\ \frac{2}{13}\sqrt{26} \end{bmatrix} \qquad (2.10)$$

```
> plv1 := plots[arrow](v1, shape=arrow,color=green,thickness=2):
  plv2 := plots[arrow](v2, shape=arrow,color=green,thickness=2):
  plv3 := plots[arrow](v3, shape=arrow,color=green,thickness=2):
  plonv1 := plots[arrow](onv1, shape=arrow,color=red,thickness=5)
  :
  plonv2 := plots[arrow](onv2, shape=arrow,color=red,thickness=5)
  :
  plonv3 := plots[arrow](onv3, shape=arrow,color=red,thickness=5)
  :
  display(plv1,plv2,plv3,plonv1,plonv2,plonv3, scaling=
  CONSTRAINED, axes=normal);
```



>

Exercises:

1) Let {(-1,1,0,0),(1,-1,1,0),(0,0,1,2)} be a basis for a certain subspace of R^4. Use the Gramm-Schmidt Process to find an orthonormal basis for this subspace.

>

Demo: Gramm-Schmidt Process

We will try to write in this section a procedure that implements the Gramm-Schmidt Process for an arbitrary list of vectors from an Euclidean space. We'll use the program to check the work that we did for the first part.

```
> myGrammSchmidt := proc(basis::list,opt::string)
  local i,j,length,temp,ogbasis,onbasis,answer:
  length:=nops(basis):
  ogbasis:=[basis[1]]:
  for i from 2 to length do
         temp := basis[i];
         for j from 1 to i-1 do
              temp:=temp-
```

```
                    DotProduct(basis[i],ogbasis[j])/
                    DotProduct(ogbasis[j],ogbasis[j])*
                    ogbasis[j]:
          od:
          ogbasis:=[op(ogbasis),temp]:
   od:
   onbasis:=[]:
   for i from 1 to length do
          temp:=1/Norm(ogbasis[i],2)*ogbasis[i]:
          onbasis:=[op(onbasis),temp]:
   od:
   if opt="orthogonal" then
          answer:=ogbasis;
   fi:
   if opt="normal" then
          answer:=onbasis;
   fi:
   answer;
   end proc:
```

We check the procedure for the basis in R^3 from part 1:

```
> l:=[v1,v2,v3];
```

$$l := \left[\begin{bmatrix}1\\1\\1\end{bmatrix}, \begin{bmatrix}-1\\3\\2\end{bmatrix}, \begin{bmatrix}-1\\0\\5\end{bmatrix}\right] \tag{3.1}$$

```
> myGrammSchmidt(l,"orthogonal");
```

$$\left[\begin{bmatrix}1\\1\\1\end{bmatrix}, \begin{bmatrix}\frac{-7}{3}\\\frac{5}{3}\\\frac{2}{3}\end{bmatrix}, \begin{bmatrix}\frac{-21}{26}\\\frac{-63}{26}\\\frac{42}{13}\end{bmatrix}\right] \tag{3.2}$$

```
> myGrammSchmidt(l,"normal");
```

$$\left[\begin{bmatrix}\frac{1}{3}\sqrt{3}\\\frac{1}{3}\sqrt{3}\\\frac{1}{3}\sqrt{3}\end{bmatrix}, \begin{bmatrix}-\frac{7}{78}\sqrt{78}\\\frac{5}{78}\sqrt{78}\\\frac{1}{39}\sqrt{78}\end{bmatrix}, \begin{bmatrix}-\frac{1}{26}\sqrt{26}\\-\frac{3}{26}\sqrt{26}\\\frac{2}{13}\sqrt{26}\end{bmatrix}\right] \tag{3.3}$$

Exercise:

2) Use the procedure myGrammShmidt and Maple's command GramSchmidt( ) to check the answers from Exercise 1.

Gramm-Schmidt Process for polynomials in P_2:

We'll try in this section to obtain an orthonormal basis for $\boldsymbol{P}_2$
(the vector space of polynomials of degree less or equal than 2), where the inner product is defined by $(p, q) = \int_{-1}^{1} p(x) q(x) \, dx$. We start with the initial basis $\{1, x, x^2\}$.

```
> p1:=x->1:
  p2:=x->x:
  p3:=x->x^2:
```

We orthogonalize first this set:

```
> ogp1:=x->p1(x):
  ogp2:=x->eval(p2(x)-int(p2(x)*ogp1(x),x=-1..1)/
                    int(ogp1(x)*ogp1(x),x=-1..1)*
                    ogp1(x)):
  ogp3:=x->p3(x)-int(p3(x)*ogp1(x),x=-1..1)/
                    int(ogp1(x)*ogp1(x),x=-1..1)*
                    ogp1(x)
                    -int(p3(x)*ogp2(x),x=-1..1)/
                    int(ogp2(x)*ogp2(x),x=-1..1)*
                    ogp2(x):
> ogp1(x);
  ogp2(x);
  ogp3(x);
```

$$1$$

$$x$$

$$x^2 - \frac{1}{3} \tag{4.1}$$

We check that they are orthogonal with respect to this inner product:

```
> int(ogp1(x)*ogp2(x),x=-1..1);
  int(ogp1(x)*ogp3(x),x=-1..1);
  int(ogp2(x)*ogp3(x),x=-1..1);
```

$$0$$

$$0$$

$$0 \tag{4.2}$$

The last operation is orthonormalization:

```
> onp1:=x->1/sqrt(int(ogp1(x)*ogp1(x),x=-1..1))*ogp1(x):
  onp2:=x->1/sqrt(int(ogp2(x)*ogp2(x),x=-1..1))*ogp2(x):
  onp3:=x->1/sqrt(int(ogp3(x)*ogp3(x),x=-1..1))*ogp3(x):
> onp1(x);onp2(x);onp3(x);
```

$$\frac{1}{2}\sqrt{2}$$

$$\frac{1}{2}\sqrt{6}\, x$$

$$\frac{3}{4}\sqrt{10}\left(x^2 - \frac{1}{3}\right) \tag{4.3}$$

We check that these three vectors are orthonormal:

```
> int(onp1(x)*onp1(x),x=-1..1);
```

```
int(onp2(x)*onp2(x),x=-1..1);
int(onp3(x)*onp3(x),x=-1..1);
```

$$1$$

$$1$$

$$1 \qquad (4.4)$$

Exercise:

1) Find an orthonormal basis for P_3 starting with the basis:

$\{3, x, x^2 - x, x^3\}$.

The inner product is defined by (p, q)= $\int_{-1}^{1} p(x)q(x)dx$.

>

>